COMPLEX FLOOR REVISITED

Doug Forkes
I. P. Sharp Associates
145 King Street West
Toronto, Ontario, Canada, M5H 1J8
(416) 364-5361

Abstract

Three functions are evaluated for their suitability as the floor function in the complex domain. The functions are evaluated against abstract criteria such as symmetry and convexity, and also for suitability in practical applications. The family of functions derived from floor is developed and examined in each case.

Introduction

In 1973 E. E. McDonnell presented a paper [1] in which he proposed an extension of floor to the complex domain. His purpose was to extend the functions floor, ceiling, residue and encode. McDonnell gave several properties which he felt a 'floor' function ought to have. In this paper I compare his floor function with two other functions, as they relate to McDonnell's criteria and certain other properties.

Notation

McDonnell introduced a notation for expressing complex numbers in APL in which the representations of the real and imaginary parts were separated by the letter *I*. Thus *3I4* represented 3+4i. In this paper *J* will separate the parts. The functions *RE* and *IM* returned the real and imaginary parts of their arguments. This paper retains that convention. Furthermore the variable *I* will stand for the constant i, that is, 0*J*1.

McDonnell's Function

McDonnell's floor function, which I shall name *MF*, can be defined as:

```
    ∇R←MF Z;A;B;A1;B1;T
[1]  A←RE Z
[2]  B←IM Z
[3]  T←⌊A+B
[4]  B1←⌈0.5×T-1+A-B
[5]  A1←T-B1
[6]  R←A1+I×B1
    ∇
```

Ceiling, residue and encode functions derived from *MF* will be named *MC*, *MR*, and *ME*.



Region where 0=MF X
Figure 1

Figure 1 shows the set of points, *X*, for which *MF X* is zero. Points on the thick part of the boundary are included; points on the thin part are excluded. The set of points whose floor is *Y*, for any integer *Y*, is *X*+*Y*, that is, the diagram is shifted so that point 0 is at *Y*.

Hurwitz's Function

In 1888 A. Hurwitz defined a complex function which I shall name *HF*.

```
   ∇R←HF Z
[1] R←(⌊.5+RE Z)+I×⌊0.5+IM Z
   ∇
```

Ceiling, residue, and encode functions derived from *HF* will be named *HC*, *HR*, and *HE*.

The region for which *HF X* is zero is shown in Figure 2.



Region where 0=HF X
Figure 2

The Third Floor

A third floor function, *FF* can be defined as

```
   ∇R←FF Z
[1] R←(⌊RE Z)+I×⌊IM Z
   ∇
```

Ceiling, residue, and encode functions derived from *FF* will be named *FC*, *FR*, and *FE*.

This function is closely related to Hurwitz's function since the latter can be written as:

```
   ∇R←HF Z
[1] R←FF 0.5J0.5+Z
   ∇
```

Tesselation

A requirement for any floor function, *F*, satisfied by *MF*, *HF*, and *FF*, is that (*F X*+*K*)=*K*+*F X* for any integer *K*. This implies that if Z is the region whose floor is 0, then the region whose floor is *K* can be found by shifting Z an amount *K*. If the entire plane is mapped in this way then no two regions can overlap (otherwise *F* would be multi-valued). Similarly if *F* is to be defined everywhere, then there can be no point which is not covered by some region. This is analogous to covering the plane with repeated instances of a tile of shape Z, without rotating Z. It is this procedure which I refer to as tesselation.

Properties

I will compare these functions by showing what definitions they lead to for the various functions that may be defined in terms of floor, what identities they satisfy and so forth. Some of McDonnell's original criteria, satisfied by all three functions, have been omitted.

Ideal Properties of a Floor Function F

a) Compatibility

MF	*HF*	*FF*
yes	no	yes

If 0=*IM Z* then (*F Z*)=⌊*Z*

This obviously rules *HF* out as a serious candidate for an extension to floor.

b) Convexity

MF	*HF*	*FF*
yes	yes	yes

If *X*=*F Z*1 and *X*=*F Z*2 then *X*=*F*(*A*×*Z*1)+*B*×*Z*2 for 1=*A*+*B*; *A*≥0 and *B*≥0.

This means that if *Z*1 has the same floor as *Z*2 then so does any value between them.

c) Fractionality

MF	*HF*	*FF*
yes	yes	no

1<|*Z*-*F Z*

This property ensures that for *A*≠0 (|*A*)>|*A*|*B* for any *B*.

d) Symmetry

MF	*HF*	*FF*
no	yes	yes

If (*P*+*I*×*Q*)=*F A*+*I*×*B* then (*Q*+*I*×*P*)=*F B*+*I*×*A*

This requires that the real and imaginary parts be treated similarly (but not necessarily independently). McDonnell required a weaker property, that if *P=F A* then *(I×P)=F I×A* for *A* real.

MF fails for example:
```
      MF 1.5J1.5
2J1
```

e) Separation — *MF* no, *HF* yes, *FF* yes

If *(-F-X)=F X* then *X=F X*

Floor and ceiling *X* are to be equal if and only if *X* is an integer, not at some other points as well. *MF* fails for example:
```
      MF 0.75J0.25
1
      MC 0.75J0.25
1
```

f) Extendability — *MF* yes, *HF* yes, *FF* no

The function can be extended to quaternions. Quaternions are the four dimensional analogue of (two dimensional) complex numbers. It can be shown that the constraints of symmetry and convexity (generalized to four dimensions) applied to *FF* imply a region for *X*, such that *0=FF X*, which does not tesselate the quaternion domain.

Extended Definition of Ceiling

Ceiling is extended by the relationship

```
      ⌈X↔ -⌊-X
```

As was noted above, *MF* has the peculiarity that *Y←MC X* equals *MF X* along a line, rather than at the point *Y* alone.

HC is just *HF* itself except that the boundaries of each region are reversed with respect to inclusion. For example:
```
      HF 0.5J0.5
0
      HC 0.5J0.5
1J1
```

Extended Definition of Residue

Residue is derived from floor through the application of the following identities:

```
      0|B ↔ B
      1|B ↔ B-⌊B
      A|B ↔ A×1|B÷A for A≠0
```

Figure 1 shows the set of points, *X*, for which *X=1 MR X*. Figure 3 shows geometrically the computation of *A MR B*.



Figure 3

Tile 0 is constructed with corners at *A×1 0J1 ¯0.5J0.5 0.5J¯0.5*. The plane is tesselated with such tiles centred on *A×0.25J0.25+K* for each integer *K*. *B* is located in some tile which is translated to overlay tile 0, with *B* falling on *B'*. *B'* is *A MR B*.

Figures 4 and 5 show the similar way in which *HR* and *FR* obtain their results. Here again there is a close relationship between *HR* and *FR* since *HR* can be defined in terms of *FR* as:

```
   ∇ R←A HR B
[1] R←(A FR B +0.5J0.5×A)-0.5J0.5×A
   ∇
```

Notice that in figure 5 part of tile 0 lies outside a circle of radius |*A* about 0. This shows the failure of 'fractionality' for *FF*. In figure 3, illustrating *MR*, tile 0 lies just within that radius, while in figure 4, illustrating *HR*, tile 0 lies well within it.

Extended Definition of Encode

Encode may be defined in terms of residue. Here are several examples with encode functions *ME*, *HE*, *FE*.

```
        TEN←10 10 10
        CTEN←10J10 10J10 10J10
        EG←789J987

        TEN ME EG
8J¯1 ¯1J8 ¯1J7
        CTEN ME EG
¯5J6 ¯2 9J7
```



B'=A HR B
Figure 4



B'=A FR B
Figure 5

```
      TEN HE EG
¯2 ¯1J¯1 ¯1J¯3
      CTEN HE EG
5J¯4 ¯1 ¯1J¯3

      TEN FE EG
7J9 8J8 9J7
      CTEN FE EG
¯6J6 8J9 ¯1J17
```

It is possible to show that for sufficiently large *K* (a function of *B*) *B*=*A*⊥(*K*⍴*A*)*HE B* provided 2<|*A*. Whether convergence criteria can be found for *ME* and *FE* remains an open question. Certainly for positive real *A*, neither function converges for negative real *B*.

Practical Applications

a) Rounding

It is frequently necessary to round real numbers to a specific number of decimal places. Presumably complex applications will require something similar. Rounding to 0 decimal places (i.e., to the nearest integer) is trivial with *HF*:

```
    ∇ R←HRND X
[1]  R←HF X
    ∇
```

and almost as trivial with *FF*:

```
    ∇ R←FRND X
[1]  R←FF 0.5J0.5+X
    ∇
```

The analogue of this function based on *MF* is:

```
    ∇ R←MRND X
[1]  R←MF 0.25J0.25+X
    ∇
```

This function does not round to the nearest integer in every case however. For example:

```
      MRND 0.3J0.4
0J1
      HF 0.3J0.4
0
```

b) Computation of GCD

McDonnell cited finding the GCD of two numbers as a practical use of residue and suggested the following function:

```
    ∇ X←W GCD Y
[1]  X←W
[2]  W←W RES Y
[3]  Y←X
[4]  →W≠0
    ∇
```

Provided *RES* is a residue function derived from a floor function with 'fractionality', the above function will converge to a GCD of its arguments. For real arguments, all versions of *GCD* converge rapidly (proportional to ⍟*W*⌊*Y* for positive integers). For complex arguments, however, substituting *FR* for *RES*, *GCD* does not always converge (e.g. 4 *GCD* 3J2). Substituting *MR* for *RES*, *GCD* always converges, but can do so in time proportional to the magnitude of its arguments (e.g. 99J1 *GCD* ¯99J¯2 takes 100 iterations). Substituting *HR* for *RES*, *GCD* continues to converge logarithmically.

c) Computation of Continued Fractions

A. Hurwitz[2] and J. O. Shallit[3] have both given algorithms for computing continued fraction expressions for complex values. Hurwitz's algorithm was based on *HF*; Shallit's was based on *MF*. In his paper Shallit gives a rule for deciding whether or not a particular continued fraction expression can be generated by his algorithm. He conjectures that no similar rule exists for Hurwitz's algorithm.

Conclusion

No complex floor function is perfect. *HF* fails to be compatible with the present floor function. *FF* fails to possess 'fractionality' and cannot be extended to the quaternions. *MF* fails to have symmetry or separation.

My feeling is that attempting to choose a floor function by satisfying identities puts the cart before the horse. The point of insisting on identities is to ensure a useful function. Nevertheless, the acid test of a primitive must be its utility - its ability to combine with the other primitives to form useful functions.

On this basis, I prefer *FF*. It can be used to find the integer parts of a complex number, or the nearest integer to a complex number (by using it to express *HF*). A GCD function written using *FF* (by substituting the *FR* expression for *HR* into *GCD*) converges faster than any GCD function written with *MR*. Similarly continued fractions found with Hurwtiz's algorithm (easily expressed with *FF*) require fewer terms for the same degree of accuracy than those found with Shallit's algorithm. The classic test for being integral, (⌊*X*)=⌈*X*, works when *FF* and *FC* are used, but not when *MF* and *MC* are (of course *X*=⌊*X* fails equally well under both definitions when *X* is near 0).

The difficulty in extending these functions can probably be traced to the definition of floor *X*: the largest of the integers less than or equal to *X*. The complex domain is not ordered, so any attempt to extend such a definition is bound to be imperfect. Hurwitz's function which computes the nearest integer does not suffer from this problem however. It must resolve equal distances in some arbitrary fashion, but it has this problem even on the reals. There is a direct analogy with the arbitrary choice for the phase angle of ¯1.

My own recommendation is that *FF* be adopted for complex floor and that if and when APL is extended to the quaternions Hurwitz's functions be adopted as primitives also. This latter recommendation applies even if *MF* is eventually chosen as the definition of complex floor, since there would otherwise be no convenient expression for it.

References

[1] McDonnell, E. E., **Complex Floor**, APL Congress 73, North-Holland, Amsterdam, 1973

[2] Hurwitz, A., **Uber die Entwicklung Complexer Grossen in Kettenbruche**, Acta Mathematica, 11, 1888

[3] Shallit, J. O., **Integer Functions and Continued Fractions**, Princeton University, 1979